Using AMPL/CPLEX to Model and Solve the Electric Vehicle Routing Problem (EVRP) with Heterogeneous Mixed Fleet

Lucas de Souza Silva
Alessandro D'Angelo

Problema VRP

- O VRP é um problema que consiste em definir rotas para um conjunto de veículos estacionados em um depósito central que irão servir um conjunto de clientes, minimizando os custos de transporte.

- Os problemas reais são caracterizados por diversas restrições que limitam a tipologia dos ciclos e tornam o VRP um dos mais difíceis entre os problemas de otimização combinatorial.

- O VRP é tido com NP-hard e os algoritmos exatos propostos na literatura são capazes de resolver apenas problemas de menores dimensões e aparentemente não levam em consideração a complexidade de problemas reais.

Função objetivo VRP

A função objetivo é minimizar a soma da distância percorrida durante toda a viagem, denotada pela distância total (Total_dis).

$$\min. \quad Total\_dis = \sum_{h \in V} \sum_{i,j \in N, i \neq j} x_{ijh} \cdot D_{ij}$$

Parâmetros VRP

Parâmetros:

N → conjunto de nós, n=cartão(N)

I → índice de cliente que i $\in$ N

V → conjunto de veículo

Dij → a distância entre o nó i e o nó j

M → Um grande número

Variáveis VRP

Variável:

xijh → Variável binária que indica se os nós i e j estão conectados.

ui → Variável inteira não negativa indicando a sequência de visitação.

Este modelo assume que a demanda do nó é zero e a capacidade do veículo é ilimitada. No entanto, o problema VRP existe amplamente em uma variedade de campos sempre relacionados à capacidade específica.

Restrições EVRP

- Considera a demanda do nó cliente e a capacidade do veículo.

- Define um novo conjunto de tipo de veículo e também exige que cada veículo seja limitado por sua capacidade.

Restrições EVRP

Novos fatores premeditados:

- Definição 1 → Phi: Variável não negativa que indica a potência restante do veículo h que está no nó i.
- Definição 2 → Ri: Parâmetro binário que significa se o nó i é um estação de carga, 1 significa que i é uma estação de carga e 0 caso contrário.
- Definição 3 → ai: Parâmetro denota a demanda do nó cliente i.

Restrições EVRP

Novos fatores premeditados:

- Definição 4 → Ch: Parâmetro denota a capacidade do veículo h.
- Definição 5 → Lh: O parâmetro mostra a distância mais longa percorrida pelo veículo h completamente carregado.

Com base nessas definições e no VRP clássico, podemos construir um problema de roteamento de veículo elétrico (EVRP).

Restrições EVRP

Constraints:

$$\sum_{h \in V} \sum_{j \in N, i \neq j} x_{jih} = 1 \qquad \forall i \in N, i > 0; R_i = 0$$

$$\sum_{h \in V} \sum_{j \in N, j \neq i} x_{ijh} = 1 \qquad \forall i \in N, i > 0; R_i = 0$$

- a primeira e a segunda restrições mostram o princípio de visita de cada nó, o que significa que se i não for um posto de carga, os veículos entram no nó i e saem do nó i uma vez.

$$\sum_{i \in N; i > 0} x_{0ih} \leq 1 \qquad \forall h \in V$$

- A terceira restrição indica que todos os veículos que fazem parte da atividade logística partem do depósito. Essa restrição garante que o roteamento de cada veículo constitui um círculo.

Restrições EVRP

$$\sum_{j \in N, j \neq i} x_{jih} = \sum_{j \in N, j \neq i} x_{ijh} \qquad \forall i \in N, h \in V$$

- A quarta restrição expressa o balanço de entrada e saída de cada nó.

$$c_h \geq \sum_{i, j \in N, j \neq i, j > 0} x_{ijh} \cdot a_j \qquad \forall h \in V; R_j = 0$$

- A quinta restrição denota a restrição de limitação de capacidade que deve ser menor que a capacidade do veículo.

Restrições EVRP

$$p_{hi} \geq L_h \qquad \forall i \in N; h \in V; R_i = 1$$

- A sexta restrição indica que o veículo está totalmente carregado quando sai de uma estação de carregamento.

$$p_{hi} \leq L_h \qquad \forall i \in N; h \in V$$

- A sétima restrição indica que a energia restante está limitada à capacidade da bateria.

Restrições EVRP

$$u_j - u_i \geq 1 - M \cdot (1 - x_{ijh})$$
$$\forall i, j \in N, i \neq j, i > 0, j > 0; h \in V$$

- A oitava restrição significa a exigência da sequência de visitas. Se o xijh for 1, ou seja, o veículo h sai do nó i e entra no nó j, a sequência de visita de j deve ser maior que a sequência de visita de i.

$$\sum_{i,j \in N; j \neq i; j>0} x_{ijh_1} \leq= 1 + \sum_{i,j \in N; j \neq i; j>0} x_{ijh_2}$$
$$\forall h_1, h_2 \in V; h_1 \neq h_2$$

- A nona restrição denota que quaisquer sequências de visita de dois veículos a diferença não é maior que 1.

Restrições EVRP

$$p_{hi} - p_{hj} \cdot (1 - R_j) \geq D_{ij} - M \cdot (1 - x_{ijh})$$

$$\forall i, j \in N, i \neq j; h \in V$$

- A décima restrição expressa que a potência restante do veículo é suficiente para dirigir até a próxima estação de carga.

O modelo EVRP levou em consideração a limitação de capacidade e substituiu o veículo a combustível pelo veículo elétrico, o que reflete melhor as atividades logísticas de distribuição reais.

Restrições EVRP

Objective:

$$\min.\ Total\_dis = \sum_{h\in V}\sum_{i,j\in N, i\neq j} x_{ijh}\cdot D_{ij}$$

Constraints:

$$\sum_{h\in V}\sum_{j\in N, i\neq j} x_{jih} = 1 \qquad \forall i\in N, i>0; R_i = 0$$

$$\sum_{h\in V}\sum_{j\in N, j\neq i} x_{ijh} = 1 \qquad \forall i\in N, i>0; R_i = 0$$

$$\sum_{i\in N; i>0} x_{0ih} \leq 1 \qquad \forall h\in V$$

$$\sum_{j\in N, j\neq i} x_{jih} = \sum_{j\in N, j\neq i} x_{ijh} \qquad \forall i\in N, h\in V$$

$$c_h \geq \sum_{i,j\in N, j\neq i, j>0} x_{ijh}\cdot a_j \qquad \forall h\in V; R_j = 0 \qquad (1)$$

$$p_{hi} \geq L_h \qquad \forall i\in N; h\in V; R_i = 1$$

$$p_{hi} \leq L_h \qquad \forall i\in N; h\in V$$

$$u_j - u_i \geq 1 - M\cdot(1-x_{ijh})$$

$$\forall i,j\in N, i\neq j, i>0, j>0; h\in V$$

$$\sum_{i,j\in N; j\neq i; j>0} x_{ijh_1} \leq = 1 + \sum_{i,j\in N; j\neq i; j>0} x_{ijh_2}$$

$$\forall h_1, h_2 \in V; h_1 \neq h_2$$

$$p_{hi} - p_{hj}\cdot\left(1-R_j\right) \geq D_{ij} - M\cdot(1-x_{ijh})$$

PARTE 2

Using AMPL/CPLEX to Model and Solve the Electric Vehicle Routing Problem (EVRP) with Heterogeneous Mixed Fleet

Lucas de Souza Silva
Alessandro D'Angelo

Relembrando

- O EVRP é um problema que consiste em definir rotas para um conjunto de veículos estacionados em um depósito central que irão servir um conjunto de clientes, minimizando os custos de transporte, sendo os veículos elétricos e portanto tendo limitações de quilometragem para carregar.

Mini-Mundo

- Teremos 21 nós, sendo 11 de demanda, 9 de posto de carga e 1 de depósito.
- Teremos 3 tipos de veículos com lotação de 200 unidades e capacidades de 0.6, 0.7, 0.8 respectivamente.

Modelo - Parâmetros

```
set N:={0..20};    # Conjunto de nós (clientes e estações de recarga)
set V:={1..3};    # Conjunto de veículos

param M = 100;
param D{N,N};  # Distância entre os nós
param Ri{N};   # Indicador de estação de recarga (1 ou 0)
param Ai{N};   # Demanda dos nós
param Ch{V};   # Capacidade de carga dos veículos
param Lh{V};   # Distância máxima percorrida pelos veículos
```

Modelo - Variáveis de Decisão

```
var x{N,N,V} binary;  # Variável binária: 1 se o veículo h vai do nó i ao nó j
var u{N,V} integer >= 0;  # Variável inteira: sequência de visita dos nós
var phi{V,N} >=0; # Varável para saber se tem combustivel ou não
```

Dados de entrada

- Desta tabela obtivemos nossa matriz D[i,j] que é a distância entre o nó i e j.
- Fórmula utilizada: $d_{AB}^2 = (x_B - x_A)^2 + (y_B - y_A)^2$

nó	X	Y
0	0.5	0.5
1	0.590166	0.609209
2	0.203093	0.189873
3	0.253639	0.921892
4	0.532339	0.957156
5	0.166748	0.105726
6	0.027142	0.714106
7	0.332709	0.551532
8	0.805947	0.263135
9	0.640516	0.349604
10	0.754375	0.407247
11	0.486731	0.665212
12	0.039934	0.575807
13	0.65443	0.942022
14	0.478125	0.363525
15	0.312899	0.003089
16	0.60368	0.755598
17	0.390284	0.450103
18	0.957052	0.170122
19	0.592204	0.787748
20	0.760299	0.837808

Dados de entrada

- 1, Posto de carga e/ou Depósito central
  0, Nó de demanda

```
param Ri :=
0 1
1 0
2 1
3 0
4 0
5 0
6 1
7 0
8 1
9 1
10 0
11 0
12 1
13 0
14 1
15 0
16 1
17 0
18 0
19 0
20 1;
```

Dados de entrada

- O parâmetro Ai é a demanda necessária em cada nó que não é de recarga.

```
param Ai :=
0 0
1 68
2 0
3 27
4 47
5 63
6 0
7 49
8 0
9 0
10 68
11 23
12 0
13 62
14 0
15 57
16 0
17 17
18 76
19 27
20 0;
```

Dados de entrada

- Os parâmetros Ch e Lh são respectivamente capacidade do veículo e distância que o veículo h pode percorrer até que a energia acabe.
- Neste caso, temos 3 carros.

```
param Ch :=
1 200
2 200
3 200;

param Lh :=
1 0.6
2 0.7
3 0.8;
```

Modelo - Função Objetiva

Objective:

$$\min.\ \ Total\_dis = \sum_{h \in V} \sum_{i,j \in N, i \neq j} x_{ijh} \cdot D_{ij}$$

```
minimize TotalDistance:
    sum{h in V, i in N, j in N: i != j} (D[i,j] * x[i,j,h]);
```

Modelo - Restrições

Constraints:

$$\sum_{h\in V}\sum_{j\in N, i\neq j} x_{jih} = 1 \qquad \forall i \in N, i > 0; R_i = 0$$

$$\sum_{h\in V}\sum_{j\in N, j\neq i} x_{ijh} = 1 \qquad \forall i \in N, i > 0; R_i = 0$$

```
subject to VisitOnce{i in N diff {0}: Ri[i] = 0}:
    sum{h in V, j in N: i != j} x[j,i,h] = 1;

subject to VisitOnce1{i in N diff {0}: Ri[i] = 0}:
    sum{h in V, j in N: i != j} x[i,j,h] = 1;
```

Modelo - Restrições

$$\sum_{i \in N; i>0} x_{0ih} \leq 1 \qquad \forall h \in V$$

$$\sum_{j \in N, j \neq i} x_{jih} = \sum_{j \in N, j \neq i} x_{ijh} \qquad \forall i \in N, h \in V$$

$$c_h \geq \sum_{i, j \in N, j \neq i, j>0} x_{ijh} \cdot a_j \qquad \forall h \in V; R_j = 0 \quad (1)$$

```
subject to AllVehiclesLeaveDepot{h in V}:
    sum{i in N diff {0}} x[0,i,h] <= 1;

subject to InAndOut{i in N, h in V}:
    sum{j in N: i != j} x[j,i,h] = sum{j in N: i != j} x[i,j,h];

subject to CapacityLimit{h in V, j in N diff {0}}:
    sum{i in N: i != j} Ai[j] * x[i,j,h] <= Ch[h];
```

Modelo - Restrições

$$p_{hi} \geq L_h \qquad \forall i \in N; h \in V; R_i = 1$$

$$p_{hi} \leq L_h \qquad \forall i \in N; h \in V$$

$$u_j - u_i \geq 1 - M \cdot (1 - x_{ijh})$$

$$\forall i, j \in N, i \neq j, i > 0, j > 0; h \in V$$

```
subject to FullyChargedOnChargeStation{h in V,i in N: Ri[i] = 1}:
    phi[h, i] >= Lh[h];

subject to RemainingChargeAfterCustomer1{h in V, i in N}:
    phi[h, i] <= Lh [h];

subject to SequenceConstraint{h in V, i in N diff {0}, j in N diff {i}: j>0}:
    u[j,h] - u[i,h] >= 1 - M * (1 - x[i,j,h]);
```

Modelo - Restrições

$$\sum_{i,j\in N; j\neq i; j>0} x_{ijh_1} <= 1 + \sum_{i,j\in N; j\neq i; j>0} x_{ijh_2}$$

$$\forall h_1, h_2 \in V; h_1 \neq h_2$$

$$p_{hi} - p_{hj} \cdot (1 - R_j) \geq D_{ij} - M \cdot (1 - x_{ijh})$$

```
subject to DiferenceSequence{h1 in V, h2 in V diff {h1}}:
    sum{i in N, j in N diff {i}: j>0} x[i,j,h1] <= 1 + sum{i in N, j in N diff {i}: j>0} x[i,j,h2];

subject to RemainingChargeAfterCustomer{i in N, j in N diff {i}, h in V}:
    phi[h,i] - phi[h,j] * (1- Ri[j]) >= D[i,j] - M *(1 - x[i,j,h]);
```

Resultados obtidos

- O menor valor encontrado pelo minimização da função objetivo foi de $\mathbf{5.376345}$ enquanto que no trabalho base foi de $\mathbf{5.0692}$.

- Acreditamos que esta diferença seja causada pela aproximação feita ao calcular os valores na matriz D além da utilização de um solver diferente.

- A sequência de visitação dos nós foi:
    - Veículo 1: $0 \rightarrow 12 \rightarrow 9 \rightarrow 10 \rightarrow 18 \rightarrow 8 \rightarrow 1 \rightarrow 0$ (144)
    - Veículo 2: $0 \rightarrow 4 \rightarrow 19 \rightarrow 9 \rightarrow 17 \rightarrow 15 \rightarrow 2 \rightarrow 5 \rightarrow 0$ (191)
    - Veículo 3: $0 \rightarrow 9 \rightarrow 11 \rightarrow 3 \rightarrow 16 \rightarrow 13 \rightarrow 6 \rightarrow 7 \rightarrow 0$ (161)
    - Nós não visitados: 14, 20 (nós de recarga)

Resultados obtidos

É possível observar na imagem ao lado que as restrições de carga das baterias foram respeitadas. Em nenhum momento o valor ultrapassou o limite estabelecido nos dados de entrada, e em todos os postos de recarga pelos quais o veículo passou, o valor atingiu o máximo permitido.

```
phi [*,*] (tr)
:    1          2          3         :=
0    0.6        0.7        0.8
1    0.149792   0          0
2    0.6        0.7        0.8
3    0          0          0.348174
4    0          0.560157   0
5    0          0.352819   0
6    0.6        0.7        0.8
7    0          0          0.251591
8    0.6        0.7        0.8
9    0.6        0.7        0.8
10   0.447015   0          0
11   0          0          0.708815
12   0.6        0.7        0.8
13   0          0          0.409042
14   0.6        0.7        0.8
15   0          0.244697   0
16   0.6        0.7        0.8
17   0          0.568752   0
18   0.150029   0          0
19   0          0.283086   0
20   0.6        0.7        0.8
;
```